

# NAG Fortran Library Routine Document

## F03AEF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F03AEF computes a Cholesky factorization of a real symmetric positive-definite matrix, and evaluates the determinant.

### 2 Specification

```
SUBROUTINE F03AEF(N, A, IA, P, D1, ID, IFAIL)
INTEGER          N, IA, ID, IFAIL
real           A(IA,*), P(*), D1
```

### 3 Description

This routine computes the Cholesky factorization of a real symmetric positive-definite matrix  $A = LL^T$  where  $L$  is lower triangular. The determinant is the product of the squares of the diagonal elements of  $L$ .

### 4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer-Verlag

### 5 Parameters

- 1: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 2: A(IA,\*) – **real** array *Input/Output*  
**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .  
*On entry:* the upper triangle of the  $n$  by  $n$  positive-definite symmetric matrix  $A$ . The elements of the array below the diagonal need not be set.  
*On exit:* the sub-diagonal elements of the lower triangular matrix  $L$ . The upper triangle of  $A$  is unchanged.
- 3: IA – INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F03AEF is called.  
*Constraint:*  $IA \geq \max(1, N)$ .
- 4: P(\*) – **real** array *Output*  
**Note:** the dimension of the array  $P$  must be at least  $\max(1, N)$ .  
*On exit:* the reciprocals of the diagonal elements of  $L$ .

5: D1 – *real* *Output*  
 6: ID – INTEGER *Output*

*On exit:* the determinant of  $A$  is given by  $D1 \times 2.0^{ID}$ . It is given in this form to avoid overflow or underflow.

7: IFAIL – INTEGER *Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

The matrix  $A$  is not positive-definite, possibly due to rounding errors. The factorization could not be completed. D1 and ID are set to zero.

IFAIL = 2

On entry,  $N < 0$ ,  
 or  $IA < \max(1, N)$ .

## 7 Accuracy

The accuracy of the determinant depends on the conditioning of the original matrix. For a detailed error analysis see page 25 of Wilkinson and Reinsch (1971).

## 8 Further Comments

The time taken by the routine is approximately proportional to  $n^3$ .

## 9 Example

To compute a Cholesky factorization and calculate the determinant of the real symmetric positive-definite matrix

$$\begin{pmatrix} 6 & 7 & 6 & 5 \\ 7 & 11 & 8 & 7 \\ 6 & 8 & 11 & 9 \\ 5 & 7 & 9 & 11 \end{pmatrix}.$$

## 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F03AEF Example Program Text
*      Mark 15 Revised.  NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NMAX, IA
      real             TWO
      PARAMETER        (NMAX=8, IA=NMAX, TWO=2.0e0)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5, NOUT=6)
*      .. Local Scalars ..
      real             D1, DETERM
      INTEGER          I, ID, IFAIL, J, N
*      .. Local Arrays ..
      real             A(IA,NMAX), P(NMAX)
*      .. External Subroutines ..
      EXTERNAL         F03AEF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F03AEF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      WRITE (NOUT,*)
      IF (N.GE.0 .AND. N.LE.NMAX) THEN
        READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
        IFAIL = 0
*
        CALL F03AEF(N,A,IA,P,D1,ID,IFAIL)
*
        WRITE (NOUT,*) 'Array A after factorization'
        DO 20 I = 1, N
          WRITE (NOUT,99998) (A(I,J),J=1,N)
20      CONTINUE
        WRITE (NOUT,*)
        WRITE (NOUT,*) 'Array P'
        WRITE (NOUT,99998) (P(I),I=1,N)
        WRITE (NOUT,*)
        WRITE (NOUT,99997) 'D1 = ', D1, '      ID = ', ID
        DETERM = D1*TWO**ID
        WRITE (NOUT,*)
        WRITE (NOUT,99997) 'Value of determinant = ', DETERM
      ELSE
        WRITE (NOUT,99999) 'N is out of range: N = ', N
      END IF
      STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,8F9.4)
99997 FORMAT (1X,A,F9.4,A,I2)
      END

```

## 9.2 Program Data

F03AEF Example Program Data

```

4
6      7      6      5
7     11      8      7
6      8     11      9
5      7      9     11

```

### 9.3 Program Results

F03AEF Example Program Results

Array A after factorization

6.0000	7.0000	6.0000	5.0000
2.8577	11.0000	8.0000	7.0000
2.4495	0.5941	11.0000	9.0000
2.0412	0.6931	1.6645	11.0000

Array P

0.4082	0.5941	0.4639	0.5283
--------	--------	--------	--------

D1 = 0.0691 ID = 12

Value of determinant = 283.0000

---